TEORÍA DE CONTROL

Introducción a Scilab

Sistemas de ecuaciones

1) Calcule las soluciones para las siguientes ecuaciones:

a) $2e^{-3t}=1$

b) $\begin{cases} 2x^2+3y^2=32 \\ -5x^2+4y=-48 \end{cases}$

c) $\begin{cases} (x+y)(x-y)=9 \\ 3x+4y=0 \end{cases}$

d) $\begin{cases} xy=-3 \\ x^2+2y^2=19 \end{cases}$

```
// Ejemplo de resolución de ecuaciones no lineales
//
clear
clc
deff('y=f(x)','y=-1+2*exp(-3*x)');
funcprot(0);
fsolve(0,f)
t=[0:0.01:2];
f1=f(t);
plot2d(t,f1);
xgrid();

clear
clc
deff('[y]=g(x)','y=[2*x(1)^2+3*x(2)^2-32,(-5)*x(1)^2+4*x(2)+48]');
[x,v]=fsolve([0,10],g,1d-10)
t=-5:0.1:5;
u=-sqrt((32-2*t^2)/3);
v=sqrt((32-2*t^2)/3);
w=(5*t^2-48)/4;
plot2d(t,[u',v',w']);
xgrid();

clear
clc
deff('[y]=g(x)','y=[x(1)*x(2)+3,x(1)^2+2*x(2)^2-19]');
//[x,v]=fsolve([5,0],g) // primera solución
[x,v]=fsolve([1,5],g) // segunda solución
t=0.1:0.1:5;
u=-3./t;
v=-sqrt((19-t^2)/2);
w=sqrt((19-t^2)/2);
plot2d(t,[u',v',w']);
xgrid();


clear
clc
deff('[y]=g(x)','y=[x(1)^2+x(2)^2-9,3*x(1)+4*x(2)]');
[x,v]=fsolve([3,-3],g) // primera solución
//[x,v]=fsolve([-3,3],g) // segunda solución
t=-5:0.1:5;
u=-sqrt(9-t^2);
v=sqrt(9-t^2);
w=-3*t/4;
plot2d(t,[u',v',w']);
xgrid();
```









Integrales definidas

2) Calcule la solución para las siguientes integrales:

a) $\int_0^1 (2x-3)dx$

b) $\int_1^e x \cdot \ln(x)\, dx$

c) $\int_1^2 \left(\frac{5-x}{x^3}\right) dx$

d) $\int_0^4 \left(1-2\sqrt{x}\right)^2 dx$

```
//Ejercicio 2
// Cáculo de la integral definida
//

// Inciso a)
clear
clc
deff('y=f(x)','y=-2*x-3');
funcprot(0);
intg(0,1,f)
t=linspace(-2,2,50);
f1=f(t);
plot2d(t,f1);
title('Integral Definida');
xgrid();
t=linspace(0,1,50);
f1=f(t);
plot2d3(t,f1,color("scilab magenta2"));
```



```
// Inciso b)
clear
clc
xdel();
deff('y=f(x)','y=x*log(x)');
funcprot(0);
intg(1,%e,f)
t=linspace(0.5,3,100);
for i=1:length(t)
    f1(i)=f(t(i));
end
plot2d(t,f1);
xgrid();
title('Integral Definida');
t=linspace(1,%e,100);
for i=1:length(t)
    f1(i)=f(t(i));
end
plot2d3(t,f1,color("red"));

// Inciso c)
clear
clc
xdel();
deff('y=f(x)','y=(5-x)/x^3');
funcprot(0);
intg(1,2,f)
t=linspace(0.8,2.5,100);
for i=1:length(t)
    f1(i)=f(t(i));
end
plot2d(t,f1);
xgrid();
title('Integral Definida');
t=linspace(1,2,100);
for i=1:length(t)
    f1(i)=f(t(i));
end
plot2d3(t,f1,color("green"));

// Inciso e)
clear
clc
xdel();
deff('y=f(x)','y=(1-2*sqrt(x))^2');
funcprot(0);
intg(0,4,f)
t=linspace(0,5,100);
for i=1:length(t)
    f1(i)=f(t(i));
end
plot2d(t,f1);
xgrid();
title('Integral Definida');
t=linspace(0,4,100);
for i=1:length(t)
    f1(i)=f(t(i));
end
plot2d3(t,f1,color("scilab cyan2"));
```







3) Calcule el área de la zona blanca.





```
//Ejercicio 3
// Cáculo de la integral definida
//
// Inciso a)
clear
clc
xdel();
deff('y=f(x)','y=(x-1)^2+2');
deff('z=g(x)','z=2*x');
t=linspace(0,4,100);
for i=1:length(t)
    f1(i)=f(t(i));
    g1(i)=g(t(i));
end
plot2d(t,g1,color("red"));
plot2d(t,f1);
t=linspace(1,3,100);
for i=1:length(t)
    f1(i)=f(t(i));
    g1(i)=g(t(i));
end
plot2d3(t,g1,color("red"));
plot2d3(t,f1,color("white"));
xgrid();title('Integral Definida');
xgrid();
funcprot(0);
a=intg(1,3,g)
b=intg(1,3,f)
I=a-b
```




```
// Inciso b)
clear
clc
xdel();
deff('y=f(x)','y=4-x');
deff('z=g(x)','z=3*x^2');
t1=linspace(0,4.5,100);
for i=1:length(t1)
    f1(i)=f(t1(i));
end
plot2d(t1,f1);
t2=linspace(0,1.5,100);
for i=1:length(t2)
    g1(i)=g(t2(i));
end
plot2d(t2,g1,color("red"));
clear t1 f1
t1=linspace(1,4,50);
for i=1:length(t1)
    f1(i)=f(t1(i))
end
plot2d3(t1,f1,color("scilab green2"));
clear t2 g1
t2=linspace(0,1,20);
for i=1:length(t2)
    g1(i)=g(t2(i));
end
plot2d3(t2,g1,color("scilab green2"));
xgrid();
title('Integral Definida');
xgrid();
funcprot(0);
a=intg(1,4,f)
b=intg(0,1,g)
I=a+b
```



Ecuaciones diferenciales

4) Calcule la solución para los siguientes sistemas de ecuaciones diferenciales y grafique su comportamiento en el intervalo indicado.

a) $\dfrac{d^2x}{dt^2}+5\dfrac{dx}{dt}+16x=5$ con $x(0)=3$ y $\dfrac{dx}{dt}(0)=0$; entre t=0 y t=10.

Se debe ingresar la ecuación diferencial de segundo orden como un sistema de dos ecuaciones de primer orden:

$$y(1)=x \;\; ; \;\; \frac{dx}{dt}=y(2) \; ; \; \frac{d^2x}{dt^2}=5-5y(2)+16y(1)$$

```
// Ejercicio 4
// Inciso a)
//
clear
clc
function [dydx]=fty(t, y),dydx=[y(2);5-5*y(2)-16*y(1)],endfunction
t0=0;
y0=[3;0];
tf=10;
t=linspace(t0,tf);
y=ode(y0,t0,t,fty);
delete();
plot2d(t',y')
xgrid();
legend("x(t)","dx/dt(t)")
title('Resolución de ecuación diferencial')
```



b) $\dot{x} = e^t - 2t^2$ con $x(0) = -1$; entre t=0 y t=5

```
// Inciso b)
//
clear
clc
function ydot=f(t, y)
    ydot=exp(t)-2*t^2
endfunction
x0=-1;
t0=0;
t=0:0.1:4;
x = ode(x0,t0,t,f);
plot2d(t,x)
xgrid();
legend("x(t)")
title('Resolución de ecuación diferencial')
```



c) $\left(x^2+9\right)\dfrac{dy}{dx} + xy = 0$ con $y(0) = 1$; entre x=0 y x=10

```
// Inciso c)
//
clear
clc
function ydot=f(x, y)
    ydot=-x*y/(x^2+9)
endfunction
y0=1;
t0=0;
t=0:0.1:10;
y = ode(y0,t0,t,f);
```



```
plot2d(t,y,color("red"))
xgrid();
legend("y(t)")
title('Resolución de ecuación diferencial')
```

d) $\dot{x} = \begin{bmatrix} 3 & 5 \\ -5 & 3 \end{bmatrix} x + \begin{bmatrix} e^{-t} \\ 0 \end{bmatrix}$ con $x(0) = \begin{bmatrix} 0 \\ 1 \end{bmatrix}$

```
// Inciso d)
//
clear
clc
function [dydt]=fty(t, x),dydt=[3*x(1)+5*x(2)+exp(-t);-5*x(1)+3*x(2)],endfunction
t0=0;
y0=[0;1];
tf=3;
t=linspace(t0,tf);
y=ode(y0,t0,t,fty);
delete();
plot2d(t',y')
xgrid();
legend("x1(t)","x2(t)")
title('Resolución de ecuación diferencial')
```



e) El oscilador de Rössler es un sistema de tres ecuaciones diferenciales ordinarias no lineales estudiadas por Otto E. Rössler. Estas ecuaciones diferenciales definen un sistema dinámico de tiempo-continuo que muestra dinámicas caóticas asociadas con las propiedades fractales del atractor.



$$\begin{bmatrix} \dot{x} \\ \dot{y} \\ \dot{z} \end{bmatrix} = \begin{bmatrix} -y-z \\ x+a\cdot y \\ b+z(x-c) \end{bmatrix}$$ con a,b y c constantes

Con a=0.2, b=0.2 y c=5.7, el sistema de ecuaciones presenta un comportamiento caótico.
Encuentre la solución del sistema de ecuaciones considerando condiciones iniciales x(0)=1, y(0)=1 y z(0)=1.
La respuesta graficada en un espacio (x,y,z) tiene la apariencia mostrada en la figura. Dibuje en base a la solución encontrada una gráfica similar a la presentada.

```
// Sistema de Rossler
// El sistema presenta un comportamiento caótico
clc;
clear; // limpiamos la ventana de comandos. Limpiamos el workspace
//
//modelo de rossler
//
function [dydx]=Rossler(t, y),
    a=0.2;      //parámetros del modelo
    b=0.2;
    c=5.7;
    dydx=[-y(2)-y(3);y(1)+a*y(2);b+y(3)*(y(1)-c)]
    endfunction
t0=0;
y0=[1;1;1];
tf=300;
t=linspace(t0,tf,5000);
y=ode(y0,t0,t,Rossler);
delete();
plot2d(t',y')
xgrid();
legend("[x(t)","y(t)","z(t)")
clf();
comet3d(y(1,:),y(2,:),y(3,:),"colors",5)
xgrid();
title("Oscilador de Rossler con a=0.2 b=0.2 y c=5.7");
xlabel("Eje X");
ylabel("Eje Y");
zlabel("Eje Z");
```





<u>Sistemas Lineales</u>

5) Dada las siguientes funciones de transferencia, calcule y grafique la respuesta al escalón y a la rampa.

a) $G(s)=\dfrac{10}{s^2+1.5s+10}$

b) $G(s)=\dfrac{(s+2)}{2(s+1)(s+10)}$

c) $G(z)=\dfrac{0.03338(z-0.8198)}{(z-0.9048)(z-0.3679)}$

```
// SISTEMAS LINEALES
// Ejercicio n°5
// Inciso a)
//
//TRANSFERENCIA
s=%s;
G1=syslin("c",10/(s^2+1.5*s+10))
// SIMULACIÓN
T = 0:0.01:5;
// Generación de rampa
u=zeros(1,length(T));
for i=1:length(T)
    u(i)=T(i);
end
// Simulación escalón
y=csim('step',T,G1);
plot2d(T',y');
// Simulación rampa
y=csim(u,T,G1);
plot2d(T',y',color("red"));
xgrid;
legend("Escalón","Rampa")
title("Respuesta 10/(s^2+1.5*s+10)")
//
```




```
// Inciso b)
//
//TRANSFERENCIA
s=%s;
G1=syslin("c",(s+2)/(2*(s+1)*(s+10)))
// SIMULACIÓN
T = 0:0.01:5;
// Generación de rampa
//u=zeros(1,length(T))
for i=1:length(T)
    u(i)=T(i);
end
// Simulación escalón
y=csim('step',T,G1);
plot2d(T',y');
// Simulación rampa
y=csim(u,T,G1);
plot2d(T',y',color("red"));
xgrid;

title("Respuesta (s+2)/(2*(s+1)*(s+10))")
```




```
// Inciso c)
//
clear
clc
//TRANSFERENCIA
z=%z;
G1=syslin("d",0.0338*(z-0.8198)/((z-0.9048)*(z-0.3679)))
P1=tf2ss(G1) // Transformación a Modelo de Estado
// SIMULACIÓN
T = 0:15;
u1=ones(1,length(T)) // Escalón
// Simulación escalón
y=dsimul(P1,u1);
plot2d2(T',y');
xgrid(color("gray"))
// Simulación rampa
// Generación de rampa
for i=1:length(T)
    u2(i)=T(i);
end
[a,b,c,d,x0]=P1(2:6)
y=c*ltitr(a,b,u2',x0);
plot2d2(T',y',color("red"));;
xgrid;
legend("Escalón","Rampa")
title("Respuesta")
```



6) Dadas las siguientes funciones de transferencia:

$$G_1(s)=\frac{10}{s(s+1)} \ ; \ G_2(s)=\frac{100(s+2)}{(s+31.6)} \ \text{ y } \ H(s)=\frac{0.1}{(s+100)}$$

Calcule las siguientes expresiones:

a) $$G_a(s)=\frac{G_1(s)G_2(s)}{1+G_1(s)G_2(s)H(s)}$$

b) $$G_b(s)=\frac{1}{H(s)}\left(\frac{G_1(s)}{1+G_1(s)G_2(s)}\right)$$

c) $$G_a(s)=\frac{\dfrac{1}{G_2(s)}}{G_2(s)+G_1(s)H(s)}$$

```
// SISTEMAS LINEALES
// Ejercicio n°6
clear
clc
//TRANSFERENCIA
s=%s;
G1=syslin("c",10,(s*(s+1)))
G2=syslin("c",100*(s+2),s+31.6)
H=syslin("c",0.1/(s+100))
// Inciso a)
Ga=G1*G2/(1+G1*G2*H)
zpk(Ga)
// Inciso b)
Gb=1/H*(G1/(1+G1*G2))
zpk(Gb)
// Inciso c)
Gc=(1/G2)/(G2+G1*H)
zpk(Gc)
```

```
--> // Inciso a)
--> Ga=G1*G2/(1+G1*G2*H)
 Ga  =

       200000 +102000s +1000s²
   ---------------------------------
   200 +3260s +3291.6s² +132.6s³ +s⁴

--> zpk(Ga)
 ans  =
                            (s+2) (s+100)
   1000 -------------------------------------------------
        (s+0.065696) (s+0.964622) (s+31.5552) (s+100.014)


--> // Inciso b)

--> Gb=1/H*(G1/(1+G1*G2))
 Gb  =

      31600 +1316s +10s²
   --------------------------
   200 +103.16s +3.26s² +0.1s³

--> zpk(Gb)
 ans  =

                  (s+31.6) (s+100)
   100 ------------------------------------
       (s+2.06495) (s^2+30.5351s+968.547)
```

```
--> // Inciso c)

--> Gc=(1/G2)/(G2+G1*H)
 Gc  =

        99856s +107174.56s² +7481.76s³ +164.2s⁴ +s⁵
   -------------------------------------------------------
   6320 +4003360s +8040100s² +5080000s³ +1050000s⁴ +10000s⁵

--> zpk(Gc)
 ans  =

                   s  (s+1)  (s+31.6)  (s+31.6)  (s+100)
   0.0001 ----------------------------------------------------
          (s+0.00158371)  (s+0.996909)  (s+2)  (s+2.00151)  (s+100)
```

7) Para los siguientes modelos de estado, calcule:

1. El polinomio característico de la planta.
2. Los autovalores de la planta.
3. La matriz de transferencia.
4. La matriz controlabilidad.
5. La matriz observabilidad.

a)

$$\begin{bmatrix} \dot{T}_c \\ \dot{T}_h \end{bmatrix} = \begin{bmatrix} -0.003 & 0.003 \\ 0.0003 & -0.0011 \end{bmatrix} \begin{bmatrix} T_c \\ T_h \end{bmatrix} + \begin{bmatrix} 0.0025 & 0 \\ 0 & 0.0008 \end{bmatrix} \begin{bmatrix} P \\ T_a \end{bmatrix}$$

$$y = \begin{bmatrix} T_c \\ T_h \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} T_c \\ T_h \end{bmatrix}$$

```
// SISTEMAS LINEALES
// Ejercicio n°7
clear
clc
// Inciso a)
//Modelo de Estados
A=[-0.003,0.003;0.0003,-0.0011]
B=[0.0025,0;0,0.008]
C=eye(2,2)
D=zeros(2,2)
P1=syslin("c",A,B,C,D)
// Polinomio Característico
Ps=poly(A,"s")
// Autovalores
z=roots(Ps)
z=spec(P1.a)
// Matriz de Transferencias
G1=ss2tf(P1)
s=%s;
G1=C*inv(s*eye(2,2)-A)*B+D
// Matriz Controlabilidad
U1=cont_mat(P1)
// Matriz Observabilidad
V1=obsv_mat(P1)
```

```
Polinomio Caracteristico
  0.0000024 +0.0041s +s²
Autovalores
  -0.0033926 + 0.i
  -0.0007074 + 0.i
Matriz de Transferencias
G1  =
     0.0000028 +0.0025s            0.000024
   ----------------------   ----------------------
   0.0000024 +0.0041s +s²   0.0000024 +0.0041s +s²

         0.0000007            0.000024 +0.008s
   ----------------------   ----------------------
   0.0000024 +0.0041s +s²   0.0000024 +0.0041s +s²

Matriz Controlabilidad
U1  =
   0.0025   0.      -0.0000075   0.000024
   0.       0.008    0.0000007  -0.0000088

Matriz Observabilidad
V1  =
   1.       0.
   0.       1.
  -0.003    0.003
   0.0003  -0.0011
```

b)

$$\begin{bmatrix} \dot{I}_A(t) \\ \dot{\omega}(t) \\ \dot{\theta}(t) \end{bmatrix} = \begin{bmatrix} -400 & -160 & 0 \\ 140 & -1 & 0 \\ 0 & 0 & 0 \end{bmatrix} \cdot \begin{bmatrix} I_A(t) \\ \omega(t) \\ \theta(t) \end{bmatrix} + \begin{bmatrix} 200 \\ 0 \\ 0 \end{bmatrix} \cdot V_R(t)$$

$$\begin{bmatrix} \omega(t) \end{bmatrix} = \begin{bmatrix} 0 & 1 & 1 \end{bmatrix} \cdot \begin{bmatrix} I_A(t) \\ \omega(t) \\ \theta(t) \end{bmatrix}$$

```
// Inciso b)
//Modelo de Estados
A=[-400,-160, 0;
   140,  -1,  0;
   0 ,   1,   0]
B=[200;
    0;
    0 ]
C=[0,1,0]
D=[0]
P1=syslin("c",A,B,C,D)
// Polinomio Característico
Ps=poly(A,"s")
// Autovalores
z=roots(Ps)
z=spec(P1.a)
// Matriz de Transferencias
G1=ss2tf(P1)
s=%s;
G1=C*inv(s*eye(3,3)-A)*B+D
// Matriz Controlabilidad
U1=cont_mat(P1)
// Matriz Observabilidad
V1=obsv_mat(P1)
```

```
Polinomio Característico
Ps  =
  22800s +401s² +s³
Autovalores
 z  =
  -332.41001 + 0.i
  -68.589993 + 0.i
   0.        + 0.i
Matriz de Transferencias
G1  =

        28000
   ---------------
   22800 +401s +s²

Matriz Controlabilidad

U1  =

   200.  -80000.   27520000.
   0.     28000.  -11228000.
   0.     0.         28000.

Matriz Observabilidad

V1  =

   0.        1.        0.
   140.     -1.        0.
  -56140.   -22399.    0.
```